ESD-TR-88-122

Project Report
CMT-111

An Improved Spherical Earth Diffraction Algorithm for SEKE

M.P. Shatz
G.H. Polychronopoulos

15 April 1988

Lincoln Laboratory

MASSACHUSETTS INSTITUTE OF TECHNOLOGY

LEXINGTON, MASSACHUSETTS

Prepared for the Department of the Air Force
under Electronic Systems Division Contract F19628-85-C-0002.

The work reported in this document was performed at Lincoln Laboratory, a center for research operated by Massachusetts Institute of Technology, with the support of the Department of the Air Force under Contract F19628-85-C-0002.

This technical report has been reviewed and is approved for publication.

FOR THE COMMANDER

Hugh L. Southall

Hugh L. Southall, Lt. Col., USAF
Chief, ESD Lincoln Laboratory Project Office

MASSACHUSETTS INSTITUTE OF TECHNOLOGY
LINCOLN LABORATORY

AN IMPROVED SPHERICAL EARTH DIFFRACTION ALGORITHM FOR SEKE

M.P. SHATZ
G.H. POLYCHRONOPOULOS
Group 46

PROJECT REPORT CMT-111

15 APRIL 1988

LEXINGTON MASSACHUSETTS

ABSTRACT

The spherical earth diffraction subroutine SPH35 in the radar propagation code SEKE has been known to cause errors in propagation loss computations for a range of combinations of antenna and target heights. In this report an efficient method to evaluate the Airy function in the complex plane is presented. This method uses the power series expansion near the origin and an integral representation elsewhere. It is more accurate and as fast as the method employed in the spherical earth diffraction subroutine SPH35 that evaluates every Airy function of Fock's series by a fourth-order polynomial fit to its logarithm. The algorithm presented was incorporated in a new spherical earth diffraction subroutine (SPH35N). It was found that, if SEKE uses this subroutine, no problems arise for normalized heights of up to 5000 (i.e. about 350 km at VHF or 17 km at K_u band).

The subroutine SPH35N, described in this report, has been used in the versions of SEKE running at Lincoln Laboratory, and is in the version of SEKE currently being supplied to other users.

TABLE OF CONTENTS

LIST OF ILLUSTRATIONS

LIST OF TABLES

1. INTRODUCTION

SEKE is a computer model that predicts the one-way propagation factor over irregular terrain by selecting, based on terrain geometry, one algorithm or a weighted combination of algorithms designed to compute specular reflection, spherical earth diffraction and multiple knife-edge diffraction losses. A detailed description of the SEKE model and computer listing of the program, can be found in Reference [1].

SEKE was designed for low antenna and target heights. It has been tested against measurements at frequencies ranging from X-band to VHF with target and antenna heights between about 10 to 1000 m [1]. When used for target and antenna heights outside this range inaccuracies and failures to produce results have been observed, mainly near and beyond geometrical horizon. These inaccuracies were caused by the spherical earth diffraction subroutine (SPH35) which uses a series due to Fock [2]. In this subroutine the values of the Airy function in Fock's series are evaluated by a fourth-order polynominal fit to their logarithms. These polynominals are tabulated for normalized effective antenna and target heights of up to 100 (i.e. 7000 m at VHF), hence limiting the applicability of SPH35 to low effective altitudes. In addition, inaccuracies in the polynomial fits result in erroneous values for the propagation factor at some combinations of antenna and target heights near the optical horizon, when many terms are needed in Fock's series.

In this report, we present a more accurate method for evaluating the Airy function in the complex plane, resulting in the extension of the range of validity of SEKE. The Airy function is evaluated using the power series expansion near the origin, and an integral representation elsewhere. The computed values of the Airy function agree to at least four digits with tabulated values.

The algorithm that evaluates the Airy function was incorporated in a new spherical earth diffraction subroutine (SPH35N). It was found that, if SEKE uses this subroutine, no problems arise for normalized heights of up to 5000, that is, about 375 km at VHF or 17 km at K_u band.

Some background information is given in Section 2 about Fock's series, and about the Airy integral and its properties. Notation is also introduced in this section. The algorithm for evaluating the Airy function is described in Section 3 by explaining the power series expansion, the Gaussian quadratures integration, and the connection formula. Certain computational difficulties resulting from the incorporation of this Airy function subroutine in the spherical earth diffraction algorithm are described, and the method used to overcome them is presented in Section 4. Finally, a comparison between predictions made by SEKE, using either SPH35 or SPH35N are provided in Section 5. A flowchart for SPH35N and its code are appended at the end of this report.

2. BACKGROUND

2.1 SEKE

Low-altitude propagation loss is influenced by atmospheric refraction and by diffraction and multipath (reflection) from the terrain over which the waves travel. As a result SEKE has to make two main decisions. It has to first decide whether to use a line-of-sight model (multipath), or a diffraction model. If the program decides to employ a diffraction model, a second decision must be made: to use multiple knife-edge diffraction or spherical earth diffraction.

The first decision is based on the clearance of the direct ray between the radar and the target. The program first locates on the terrain profile the highest mask of the minimum clearance point, M. Let Δ be the clearance of the direct ray at the point M and d_2 and d_2 be the distances from the radar to M and from M to the target respectively. The Fresnel clearance at the point M is given by the formula:

$$\Delta_0 = \sqrt{\frac{\lambda d_1 d_2}{d_1 + d_2}}$$

where λ is the wavelength. If $\Delta/\Delta_o > 1$ SEKE uses multipath alone. If $\Delta/\Delta_o < 1/2$ SEKE uses the diffraction subroutines. For the intermediate zone where $1 > \Delta/\Delta_o > 1/2$, a weighted average of the multipath loss F_M and the diffraction loss F_D is used (see Reference [1] for more detail). When $\Delta/\Delta_o < 1$ a second decision is made between multiple knife-edge and spherical earth diffraction. The key parameter in this solution is the ratio h_M/Δ_o, where h_M is the height of the highest mask or minimum clearance point M, measured from the best-fit line on the terrain profile. When the height of the discrete obstacles over smooth earth are large relative to Δ_o, knife-edge diffraction should dominate over spherical earth diffraction. In SEKE the propagation loss is approximated by knife-edge diffraction alone when $h_M/\Delta_o > 1/2$, by spherical earth diffraction when $h_M/\Delta_o < 1/4$ and a weighted average of the two is used in the intermediate region.

2.2 Inaccuracies in SEKE

Figure 1 shows some of the inaccuracies in SEKE caused by the SPH35 routine. The plot shows the two-way propagation factor over a smooth conducting spherical earth with an antenna height of 10 m and a target height of 8000 m at 167 MHz. At ranges below 185 km, SEKE uses the geometric optics routine GEOSE. After that point, SEKE attempts to use SPH35, but that routine fails. SEKE falls back in GEOSE, which remains fairly accurate until about 350 km. Then SEKE starts producing incorrect results. At the optical horizon, which is at 385 km, SEKE is still attempting to use SPH35, which still fails, but



Fig. 1. Prediction of the two way propagation loss as a function of range for antenna height = 10 m and target height = 8000 m at VHF (167) MHz). The solid line shows the prediction according to SPH35 and the dashed line according to SPH35N.

SEKE now uses the knife-edge routine LAPKE. This results in erroneous values. At a range of 620 km, SPH35 returns a value acceptable to SEKE, which is, however, incorrect. (The dashed line of Figure 1 shows the values returned by SPH35N). Further study shows that the problems in SPH35 are associated with the computation of Airy functions in that routine.

2.3 Spherical Earth Diffraction Subroutine and Fock's Series

Fock has shown [3], that the propagation factor, F, is a function of normalized antenna height, y, normalized target height, z, and normalized range, x, given by the following sum:

$$F(x,y,z) = 2\sqrt{\pi x}\sum_{n=1}^{\infty} f_n(y)f_n(z)\exp(\frac{1}{2}(\sqrt{3}+i)a_n x) \tag{1}$$

F is the propagation factor, defined here to be the ratio of the electric field at a point to the free space electric field at that point,

$$f_n(u) = \frac{Ai(a_n + \exp(\frac{\pi i}{3})u)}{\exp(\frac{\pi i}{3})Ai'(a_n)} \tag{2}$$

Ai(w) is the Airy integral: $Ai(w) = \frac{1}{\pi}\int_0^\infty \cos(\frac{1}{3}t^3 + wt)dt$

and a_n are the zeroes of the Airy integral.

An antenna height (h_a) and a target height (h_t), correspond to normalized heights of

$$y = \frac{h_a}{h_0} \qquad z = \frac{h_t}{h_0} \tag{3}$$

respectively, where the normalization constant is given by:

$$h_0 = \frac{1}{2}\left(\frac{a\lambda^2}{\pi^2}\right)^{\frac{1}{3}} \qquad (4)$$

where a is the effective radius of earth, and λ is the wavelength. The range normalization factor r_o and the normalized range x are given by:

$$r_0 = \left(\frac{a^2\lambda}{\pi}\right)^{\frac{1}{3}} \quad x = \frac{r}{r_0} \qquad (5)$$

SPH35, the algorithm that has been employed to evaluate the propagation factor in the spherical earth diffraction region, uses the first n terms ($n \geq 35$) in Fock's series. The convergence criterion is that the contribution of each one of three successive terms of the series is less than $\delta = 0.0005$. If this criterion is not met while considering the first 35 terms, SPH35 returns a message that it "diverged". When SPH35 "diverges" SEKE applies the geometrical optics subroutine if the point is visible, otherwise it uses the knife-edge diffraction subroutine.

SPH35 expresses $f_n(u)$ as $f_n(u) = \exp(\lambda_n(u) + i\mu_n(u))$, where $\lambda_n(u)$ and $\mu_n(u)$ are fitted by fourth order polynomials in u. The coefficients of the polynomials for each term of the series are tabulated in SPH35. In order to make the polynomial fit more accurate, the range of normalized heights (u) is divided in four separate regions. (See Table I). $f_n(u) = u$ has been found to be a good approximation in the interval $0 \leq u < 0.05$. For the regions, $0.05 \leq u < 0.2$, $0.2 \leq u < 3$, $3 \leq u < 10$, and $10 \leq u < 100$ four different polynomial fits are used. No attempt is made in this original version of SPH35 to obtain accurate polynomial fits to λ_n and μ_n for $u > 100$. In this high altitude region, SPH35 returns coefficients for λ_n

TABLE I

Approximations Used in SPH35 to Evaluate $f_n(u)$ for Different Ranges of Normalized Target Height u

RANGE	CALCULATION of $f_n(u)$
$0 \leq u < 0.05$	$f_n(u) = u$
$0.05 \leq u < 0.2$	1st fourth-order polynomial fit
$0.2 \leq u < 3$	2nd fourth-order polynomial fit
$3 \leq u < 10$	3rd fourth-order polynomial fit
$10 \leq u < 100$	4th fourth-order polynomial fit
$u > 100$	value of u out of range (however SPH35 returns a value for F according to the 4th fourth-order polynomial fit)

and μ_n according to the fourth polynomial fit. As a result, the accuracy of SPH35 is limited to normalized heights of up to 100 (i.e. 7000 m at VHF, or 317 m at K_u band).

There are two potential problems with SPH35. First, there is no polynomial fit for u > 100. Second, even for lower altitudes, inaccuracies in the polynomial fits result in erroneous values for the propagation factor or failure to converge for some combinations of antenna and target heights near the optical horizon. These discrepancies occur in regions where the sum of the series is much smaller than the largest term, or where many terms are required for convergence.

In order to test the accuracy of the polynomial fits, the values returned by SPH35 were compared with the values returned by another subroutine PROPSES. PROPSES calculates the Airy functions by numerically integrating the differential equations which define them. This technique results in high accuracy but requires long computational time.

Table II shows the percent error in SPH35. Sample runs are presented for each of the polynomial fits. It is seen that the fourth polynomial fit is inaccurate even far into the diffraction region. The remaining fits seem to work adequately far into the diffraction region, but SPH35 sometimes fails to return a value or return an inaccurate value when called well inside the horizon.

It is apparent from the above discussion and the sample run presented that the accuracy of the fourth order polynomial fits for $f_n(u)$ is not always sufficient to obtain a reasonable prediction for the propagation factor. As a result, there is a need for a new spherical earth diffraction subroutine where the Airy function is evaluated using a more accurate method. The remainder of this report describes such a new spherical earth diffraction subroutine.

TABLE II

SAMPLE RUNS FOR FINDING THE PERCENT ERROR OF $f_n(u)$ FOR EQUAL ANTENNA AND TARGET HEIGHTS

Method for calculating $f_n(u)$ in SPH35	$h_a = h_t$ (m)	% Error at first point where spherical earth diffraction is used by SEKE (well within the optical horizon)	% Error at a range of 1.5 times the optical horizon
$f_n(u) = u$	3	18.07	~ 0
1st fit	10	SPH35 failed	0.15
2nd fit	70	~ 0	0.03
3rd fit	600	16.68	0.72
4th fit	5,000	SPH35 failed	28.65

2.4 Properties of the Airy Function

Algorithms for the evaluation of the Airy function with real arguments are readily available [8]. However, their modification to handle complex arguments is not straightforward. The power series expansion can be used only in a small region near the origin, and the asymptotic formulas do not give sufficient accuracy for moderate values of z = x + iy [3].

The Airy function Ai(z) $Ai(z) = \frac{1}{\pi}\int_0^\infty \cos(\frac{1}{3}t^3 + zt)dt$

satisfies the differential equation

$$\frac{d^2u(z)}{dz^2} - z \cdot u(z) = 0$$

The two sets of linear independent solutions to the above equation are Ai(z), Bi(z) and Ai[(z exp $\pm$ (2πi/3)] where

$$Bi(z) = \frac{1}{\pi}\int_0^\infty [\exp(-\frac{1}{3}t^3 + zt) + \sin(\frac{1}{3}t^3 + zt)]dt \tag{6}$$

The functions are entire and so have convergent power series.

Schulten et al [4] give an integral representation for the Airy function whose evaluation by a Gaussian quadrature method requires only a few terms. The integral representation for Ai(z) is derived from an expression for the modified Bessel function of the second kind K_υ (z). (formula 6.627 of Ref.[5].

$$\int_0^\infty \frac{x^{-1/2}e^{-x}K_\nu(x)}{x+\zeta}dx = \frac{\pi e^\zeta K_\nu(\zeta)}{\zeta^{1/2}}\cos(\nu\pi) \qquad |\arg\zeta| < \pi \quad Re(\nu) < \frac{1}{2} \tag{7}$$

If we set $\upsilon = 1/3$, $\zeta = \frac{2}{3} z^{3/2}$ and substitute:

$$K_{1/3}(x) = \frac{\pi\sqrt{3}}{(\frac{3}{2}x)^{1/3}} Ai\left[\left(\frac{3x}{2}\right)^{2/3}\right] \tag{8}$$

equation (7) can be solved for Ai(z).

$$Ai(z) = \frac{1}{2}\pi^{-1/2} z^{-1/4} e^{-2/3z^{3/2}} \int_{0}^{\infty} \frac{\rho(x)}{1+\frac{3x}{2z^{3/2}}} dx \qquad |argz| < \tfrac{2\pi}{3} \; |z| > 0 \tag{9}$$

ρ(x) is a non-negative exponentially decreasing function:

$$\rho(x) = \pi^{-1/2} 2^{-11/6} 3^{-2/3} x^{-2/3} e^{-x} Ai\left[\left(\frac{3x}{2}\right)^{2/3}\right] \tag{10}$$

This integral representation is valid in the sector $|\arg z| < 2\pi/3$. However, there exists a connection formula that transforms a point outside this sector to a weighted sum of two linearly independent points inside it:

$$Ai(z) = e^{\pi i/3} Ai(ze^{-2\pi i/3}) + e^{-\pi i/3} Ai(ze^{2\pi i/3}) \tag{11}$$

(See formula 10.4.7 of Ref. [3]).

Even though ρ(x) contains the Airy function, the weights and abcissas for the Gaussian quadrature can be computed without an accurate computation of the Airy function, because the moments of ρ(x) can be evaluated in closed form.

3. ALGORITHM FOR EVALUATING THE COMPLEX AIRY FUNCTION

As mentioned above there exists a power series expansion for the Airy function. This series method is used to evaluate the value of the Airy function close to the origin. For large values of z a Gaussian quadratures method is implemented to evaluate the integral representation above. These two numerical methods are used in the part of the complex plane where $|\arg z| < 2\pi/3$. For the remaining part of the plane the connection formula is used.

It was found by looking at the values returned by these two different methods (power series and Gaussian quadratures)



Fig. 2. Regions on the complex plane where the power series expansion, the Gaussian quadratures method to solve the integral representation and the connection formula are each used.

on certain radials of the right half-plan ($x>0$) that the two methods returned the same value for Ai(z), within four significant digits, when $|z| \approx 2$. As a result, the algorithm was implemented so that if z lies in the right half-plane and $|z| \leq 2$, then Ai (z) is evaluated by the power series expansion. If $|z| > 2$ (and $x > 0$) then Ai(z) is evaluated by the Gaussian quadratures method. When $x < 0$ (left-hand plane), it was found that the two methods overlapped at $|z| \approx 4$. As a result if $|z| > 4$ and $x < 0$, then the power series expansion algorithm is used. If $|z| > 4$ and $x < 0$, the Gaussian quadratures method is applied. Figure 2 defines the region on the complex plane where the power series expansion, the Gaussian quadratures method and the connection formula are used.

3.1 Power (Taylor) Series Expansion

The Airy function is entire, so a convergent Taylor series representation exists for it (See Eq. 10.4.2 of Ref. [3]). This series converges very fast for small values of z. When a large value of z is used, the series converges very slowly, and inaccuracies occur due to large cancellations. The Taylor series expansion is given by:

$$Ai(z) = \alpha \cdot h(z) - \beta \cdot g(z) \tag{12}$$

where:

$$\alpha = Ai(0) = \frac{Bi(0)}{\sqrt{3}} = \frac{3^{-2/3}}{\Gamma(2/3)} = 0.3550280538 \tag{13}$$

$$\beta = -A'i(0) = \frac{B'i(0)}{\sqrt{3}} = \frac{3^{-1/3}}{\Gamma(1/3)} = 0.2588194037 \tag{14}$$

$$h(z) = \sum_{k=0}^{\infty} 3^k \left(\frac{1}{3}\right)_k \frac{z^{3k}}{(3k)!} = 1 + \frac{1}{3!}z^3 + \frac{1 \cdot 4}{6!}z^6 + \frac{1 \cdot 4 \cdot 7}{9!}z^9 + \cdots \tag{15}$$

$$g(z) = \sum_{k=0}^{\infty} 3^k \left(\frac{2}{3}\right)_k \frac{z^{3k+1}}{(3k+1)!} = z + \frac{2}{4!}z^4 + \frac{2 \cdot 5}{7!}z^7 + \frac{2 \cdot 5 \cdot 8}{10!}z^{10} + \cdots \tag{16}$$

$$3^k(\gamma + \frac{1}{3})_0 = 1$$

$$3^k(\gamma + \frac{1}{3})_k = (3\gamma + 1)(3\gamma + 4) \cdots (3\gamma + 3k - 2) \tag{17}$$

$$(\gamma \text{ arbitrary, k=1,2, } \cdots)$$

Equation (12) is implemented in the algorithm that evaluates the Airy Function in the power series region (see Figure 2).

In order to reduce the growth of the round-off error, the series are evaluated term by term:

$$Ai(z) = \sum_{n=1}^{m} a_n(z)$$

for an m such that $a_m(z) < 10^{-10}$, where

$$a_n(z) = \alpha\ h_n(z) - \beta\ g_n(z)$$

and $h_n(z)$ and $g_n(z)$ are the nth terms in the expressions for g(z) and h(z).

3.2 N-Point Gaussian Quadratures Approximation

As mentioned above Ai(z) is evaluated for large z by the generalized Gaussian quadratures method. The challenging part with this method is to find a function $\rho(x)$ so that the integrable singularities are removed from the Airy integral. Given $\rho(x)$ and given the number of terms N that should be used, one can find a set of weights w_i and abscissas x_i such that the approximation:

$$\int_b^a \rho(x)p(x)dx \simeq \sum_{i=1}^{N} w_i p(x_i) \tag{18}$$

is exact, if p(x) is a polynomial of degree less than 2N. Schulten et al.[4] give an integral representation for Ai(z), whose evaluation by a Gaussian quadrature method requires only a few terms:

$$Ai(z) = \frac{1}{2}\pi^{-1/2}z^{-1/4}e^{-\zeta}\int_0^\infty \frac{\rho(x)}{x+\zeta}dx$$
$$|z| > 0 \text{ and } |\arg\zeta| < \pi \tag{19}$$

where

$$\rho(x) = \pi^{-1/2}2^{-11/6}3^{-2/3}x^{-2/3}e^{-x}Ai\left[\left(\frac{3x}{2}\right)^{2/3}\right] \quad \text{and} \quad \zeta = \frac{2}{3}z^{3/2} \tag{20}$$

The integral portion of equation (19) can be approximated by the quadrature formula:

$$I(\zeta,\rho) = \int_0^\infty \frac{\rho(x)}{\zeta+x} \simeq \sum_{i=0}^{n} \frac{w_i}{\zeta+x_i} \tag{21}$$

The weights w_i and the abscissas x_i were found by implementing the procedure described by Press [6].

The "scalar product of two functions f and g over a weight function $\rho(x)$", is defined as:

$$< f|g >= \int_a^b \rho(x)f(x)g(x)dx \tag{22}$$

A set of orthogonal polynomials that includes exactly one polynomial of order j, called $p_j(x)$ j=0, 1, 2, ... is needed to find the weights and abscissas.

This set of polynomials can be constructed by the following recurrence relation:

$$p_0(x) = 1$$

$$p_{i+1}(x) = \left[x - \frac{< xp_i|p_i >}{< p_i|p_i >}\right] p_i(x) - \left[\frac{< p_i|p_i >}{< p_{i-1}|p_{i-1} >}\right] p_{i-1}(x) \tag{23}$$

(the second term is omitted when i=0), and

$$< xp_i|p_i > = \int_a^b \rho(x) x p_i(x) p_i(x) dx \tag{24}$$

The integral of equation (24) can be readily evaluated by observing that $\rho(x)$ is a solution to the Stieltjes moment problem, whose moments μ_k can be explicitly evaluated: (Formula 6.621.3, [5])

$$\mu_k = \int_0^\infty x^k \rho(x) dx = \frac{\Gamma(3k + 1/2)}{54^k k! \Gamma(k + 1/2)} \qquad \mathrm{k} = 0,1,2,\cdots \tag{25}$$

as a result $<xp_i|p_i>$ becomes a sum of μ_k's. Once the abscissas x_1, x_2 .... x_N (i.e. the N zeroes of $p_N(x)$) are known, the weights w_i can then be found.

Press [6] presents a simple method to find the w_i's. A new sequence of polynomials $\varphi(x)$ is constructed, by the following recurrence:

$$\begin{aligned} \varphi_0(x) &= 0 \\ \varphi_1(x) &= p_1' \int_a^a \rho(x) dx = 1 \qquad (26) \\ \varphi_{i+1}(x) &= \left[x - \frac{< xp_i|p_i >}{< p_i|p_i >}\right] \varphi_i(x) - \left[\frac{< p_i|p_i >}{< p_{i-1}|p_{i-1} >}\right] \varphi_{i-1}(x) \end{aligned}$$

where p'_1 is the derivative of p_1 (x).

The weights of an N-point Gaussian quadrature are given by the relation:

$$w_i = \frac{\varphi_N(x_i)}{p'_N(x_i)} \qquad \text{i=1,2,} \cdots, \text{N} \tag{27}$$

The procedure described above was implemented to find the weights w_i and abscissas x_i for the N-point Gaussian quadrature approximation. When the Airy function is evaluated by this method it is dependent on N, the number of terms used in the summation:

$$Ai(z,N) \equiv \frac{1}{2}\pi^{-1/2}z^{-1/4}e^{-\zeta}\sum_{i=0}^{N}\frac{w_i}{\zeta + x_i}$$
$$\text{where} \quad \zeta = \frac{2}{3}z^{2/3} \tag{28}$$

We calculated Ai (z, N) (for $|z| > 4$) for N =1, 2 ...20. It was found that the values of Ai(z, N) were almost the same for values of N near 10. For larger or smaller N, the values of Ai (z, N) were quite different indicating truncation or round-off error. It was thus decided, that N=10, is a good choice for the number of terms that should be used in the Gaussian quadratures approximation. Table III gives a list of the weights w_i and abscissas x_i for the 10-term Gaussian Quadrature integration for the Airy function.

4. ALGORITHM FOR EVALUATING THE PROPAGATION FACTOR IN THE SPHERICAL EARTH DIFFRACTION REGION (SPH35N)

4.1 Computational Difficulties and Solutions

The above algorithm for evaluating the Airy function was implemented in a new spherical earth diffraction subroutine (SPH35N). However, the incorporation of this algorithm to

TABLE III

10-TERM GAUSSIAN QUADRATURES INTEGRATION FOR THE AIRY FUNCTION

i	Abscissas x_i	Weights w_i
1	1.408308107197377E+01	2.677084371247434E-14
2	1.021488548060315E+01	6.636768688175870E-11
3	7.441601846833691E+00	1.758405638619854E-08
4	5.307094307915284E+00	1.371239148976848E-06
5	3.634013504378772E+00	4.435096659959217E-05
6	2.331065231384954E+00	7.155501075431907E-04
7	1.344797083139945E+00	6.488956601264211E-03
8	6.418885840366331E-01	3.644041585109798E-02
9	2.010034600905718E-01	1.439979241604145E-01
10	8.059435921534400E-03	8.123114134235980E-01

compute $f_n(u)$ in Fock's series, is not straightforward. A problem arises in the evaluation of $Ai(w)$, where $w = a_n + \exp(\pi i/3)\, u$, by the Gaussian quadrature method. As $|w|$ becomes large, in the region $|\arg w| > \pi/3$, the expression $\exp(-2/3\, w^{3/2})$, could overflow. However, looking at Fock's series the term $\exp(1/2(\sqrt{3} + i)\, a_n x)$ can be used to partially cancel a large value for $\exp(-2/3\, w^{3/2})$. The subroutine that evaluates the Airy function returns $\overline{Ai}(w) \equiv Ai(w) \exp(2/3\, w^{3/2})$, instead of $Ai(w)$. The Fock's series equation implemented in SPH35N, then becomes:

$$F(x,y,z) = 2\sqrt{\pi x}\sum_{n=1}^{\infty} \overline{f_n}(y)\overline{f_n}(z)\exp\left[-\psi - \zeta + \left(\frac{1}{2}(\sqrt{3}+i)a_n x\right)\right] \qquad (29)$$

where

$$\overline{f_n}(u) = \frac{\overline{Ai}\left(a_n + \exp(\frac{\pi i}{3})u\right)}{exp(\frac{\pi i}{3}Ai'(a_n)}\ , \quad \psi = \frac{2}{3}y^{3/2} \quad \text{and} \quad \zeta = \frac{2}{3}z^{3/2}$$

The connection formula used when $\overline{Ai}(z)$ is evaluated instead of $Ai(z)$, becomes:

$$\overline{Ai}(z) = e^{\pi i/3}\overline{Ai}\left(ze^{-2\pi i/3}\right)e^{4/3z^{3/2}} + e^{-\pi i/3}\overline{Ai}\left(ze^{2\pi i/3}\right) \qquad (30)$$

$$\text{where } \overline{Ai}(w) = Ai(w)e^{2/3w^{3/2}}$$

SPH35N considers as many terms in Eq. (29) as are needed to get two consecutive terms that contribute, in absolute value, less than 0.0005. If this does not occur within the first 35 terms, SPH35N is considered to have "diverged". One more check is performed in SPH35N to make sure that the computation is accurate. If any one of the terms in equation (29) contributes more than 10,000, SPH35N is interrupted. A maximum contribution of 10,000 per term was picked since $Ai(w)$ is evaluated with an accuracy of at least four significant digits. This criterion ensures that SPH35N does not produce wholly spurious results because of cancellation.

Empirically, 10,000 seemed to work very well. SEKE does try, for some few combinations of antenna and target heights, to use SPH35N in the multipath region far above the optical horizon. When this happens, a term in SPH35N contributes more than 10,000, making it impossible, given the accuracy of the Airy function, to cancel its contribution. Had this check not been included, SPH35N would have incorrectly concluded that the series converged. With this check, SEKE uses geometric optics, which is accurate in this region.

In certain other cases, where a large obstacle masks the target, SEKE tries to use a combination of spherical earth diffraction and knife-edge diffraction. In some of these cases, even though the target is masked, it can be well above the horizon for the smooth fit to the terrain used by SEKE to determine the effective heights for the spherical Earth routine; this sometime results in SPH35N "diverging". When this happens, SEKE is forced to use only knife-edge diffraction as expected, since the dominant effect is the obstacle.

4.2 Range of Validity of SPH35N when Called from SEKE

The range of validity of SPH35 in target-to-antenna distance x, was investigated for a fixed normalized target height, y, and antenna height, z. It was found that the range where SPH35N broke down was always in a region where SEKE would use geometrical optics, since the minimum clearance was found to be greater than a Fresnel clearance for any combination of y and z, where $y < 5000$ and $z < 5000$ (i.e. heights of less than 375 km at VHF or 17 km at K_u band). As mentioned above, in practice SEKE occasionally calls the sphere diffraction program in a region where it fails to return an answer. SEKE then either uses the geometric optics calculation when the point is in the multipath region or the knife-edge diffraction calculations when the point is masked. It has been found empirically that SEKE makes an accurate decision in such cases.

5. EVALUATION OF SPH35N

SPH35N was run for certain cases where SPH35 would not work correctly either because y or z were greater than 100, or because of cancellation error when y and z are less than 100. Two examples are shown in Figures 3 and 4. Figure 3 presents the two-way propagation factor F^2 in dB with respect to range for antenna heights h_a = 10 m (z = 0.14) and target height h_t = 2.000 m (y = 28.57) at VHF. The solid line gives the prediction using SPH35 and the dashed line the prediction using SPH35N. The spike in the prediction disappears when SPH35N is used. Figure 4 presents another plot of F^2 (dB) with respect to range for antenna height h_a = 6000 m and target height h_t = 10 m. In this case, the plot resulting by the use of SPH35 shows most of the problems created by the inaccurate evaluation of the Airy function in the spherical earth diffraction subroutine. At around 275 km, SEKE tries to use spherical earth diffraction, but SPH35 "diverges", so SEKE uses the geometrical optics subroutine up to the optical horizon. Beyond the optical horizon knife-edge diffraction is used, returning a constant value of F^2 = -6 dB. At around 350 km SPH35 "converges" and returns a value. When SPH35N is employed, the resulting curve is smooth. Problems as the ones encountered when using SPH35 do not occur, since SPH35N is able to return a value when called by SEKE, as expected from the specified range of validity of SPH35N.

6. CONCLUSIONS

We have presented a subroutine that evaluates the propagation factor in the spherical earth diffraction region by applying Fock's series using an accurate and efficient algorithm to evaluate the complex Airy function.

The Airy function is evaluated by the power series expansion for $|z|$ close to the origin, by the 10 term Gaussian



Fig. 3. Prediction of the two way propagation loss as a function of range for antenna height = 10 m and target height = 2000 m at VHF (167 MHz). The solid line shows the prediction according to SPH35 and the dashed line according to SPH35N.



Fig. 4. Prediction of the two way propagation loss as a function of range for antenna height = 10 m and target height = 6000 m at VHF (167 MHz). The solid line shows the prediction according to SPH35 and the dashed line according to SPH35N.

quadrature approximation for large $|z|$ and the connection formula for $|\arg z| < (2/3)\pi$ i.e. in the region where the integral representation is invalid. The implemented Airy function checks with Airy tables within four significant digits. Incorporating this subroutine in SEKE eliminated inaccuracies caused by the old spherical earth diffraction subroutine (SPH35), that uses a fourth-order polynomial fit to approximate the Airy function. The new subroutine, SPH35N, was found to be equally fast as SPH35 and adequately accurate for normalized heights of less than 5000, so that it performs as desired when called from SEKE.

REFERENCES

1. S. Ayasli, and M.B. Carlson, "SEKE: A Computer Model for Low-Altitude Radar Propagation Over Irregular Terrain," Project Report CMT-70, Lincoln Laboratory, MIT (1 May 1985). Most of the contents of this report can also be found in IEEE Trans. Antennas Propag. AP-34, 1013 (1986).

2. V.A. Fock, Electromagnetic Diffraction and Propagation Problems (Pergamon Press Ltd., Oxford, England, 1965).

3. M. Abramowitz, and I. A. Stegun, Handbook of Mathematical Functions, Sect. 10, (National Bureau of Standards, U. S. Govt. Printing Office, Washington, D.C., 1970).

4. D.G.M. Anderson, and R.G. Gordon and Z. Schulten, "An Algorithm for the Evaluation of the Complex Airy Functions," J. Comput. Phys. 31, 60. (1979).

5. I.S. Gradshetyn, and I. M. Ryzhik, Tables of Integrals, Series, and Products (Academic Press, New York, 1965).

6. W.H. Press, et al., Numerical Recipes, The Art of Scientific Computing (Cambridge University Press, Cambridge, England, 1986).

7. M.L. Meeks, Radar Propagation at Low Altitudes (ArtecHouse Inc., Dedham, Massachusetts, 1982).

8. ACM, Program 498 from TGMS (Transactions of Mathematical Software), "The Airy Subroutine".

APPENDIX I

FLOWCHART OF SPH35N


START
N = 0
x = r/d0
y = ha/h0
z = ht/h0
N = N+1
N = 36
YES
F = -1
RETURN
Ai (w)
w = an+ e^{πi/3} · y
Ai (u)
u = an+ e^{πi/3} · z
TERM = Ai(w) / (e^{πi/3} Ai'(an)) · Ai(u) / (e^{πi/3} Ai'(an)) · exp {-ω -ν + [½(√3 + i)an x]}
ω = 2/3 w^{3/2} ; ν = 2/3 u^{3/2}
|TERM| > 10000
YES
F = -2
RETURN
NO
SUM = SUM + TERM
|TERM| < 0.0005
AND
PREVIOUS
|TERM| < 0.0005
YES
F = 2√πx · |SUM|
RETURN
NO

APPENDIX II

FLOWCHART OF AI(Z)


START
$|\arg z| < \frac{2\pi}{3}$
YES
NO
CONNECTION FORMULA
$\overline{Ai}(z) = e^{\pi i/3}\, Ai(z')\, \exp(4/3\, z^{3/2}) + e^{-\pi i/3}\, Ai(z'')$
$z' = z\, e^{-2\pi i/3}$
$z'' = z\, e^{-4\pi i/3}$
$z = z'$
AND
$z = z''$
REAL $(z) > 0$
YES
$|z| < 2$
POWER SERIES
$\overline{Ai}(z) = [0.355\, h(z) - 0.258\, g(z)]\, \exp(\zeta)$
$h(z) = 1 + \frac{1}{3!} z^3 + \frac{1 \cdot 4}{6!} z^6 + \ldots$
$g(z) = z + \frac{2}{4!} z^4 + \frac{2 \cdot 5}{7!} z^7 + \ldots$
$\zeta = \frac{2}{3} z^{3/2}$
NO
YES
$|z| \leq 4$
RETURN
GAUSSIAN QUADRATURES
NO
$\overline{Ai}(z) = \frac{1}{2} \pi^{-1/2} z^{-1/4} \sum_{i=1}^{10} \frac{w_i}{\zeta + x_i}$
RETURN
$\zeta = \frac{2}{3} z^{3/2}$
w_i QUADRATURE WEIGHTS
x_i ABSCISSAS

APPENDIX III

LISTING OF SPH35N

Listing of SPH35N

```
C******************************************************************
C                                                                 *
C       SPHERICAL EARTH DIFFRACTION LOSS SUBROUTINE              *
C          APPLYING THE ANALYSIS  DESCRIBED IN                    *
C FOCK,V.A.,ELECTROMAGNETIC DIFFRACTION AND PROPAGATION PROBLEMS,*
C           OXFORD:PERGAMMON PRESS, LTD , 1965                    *
C                                                                 *
C           GEORGE H. POLYCHRONOPOULOS   1 /87                    *
C                                                                 *
C******************************************************************


      SUBROUTINE SPH35N (RNG,Z1,Z2,WAVE,AE,F)

      REAL RNG,Z1,Z2,WAVE,AE,F
      REAL *8 X,Y,Z,COEF1,COEF,SENSI,ATER,APTERM,DO,HO
      COMPLEX*16 COEF2,EPI,FNY,FNZ,TERM,ETERM,SUM,FAIRY,AY,AZ,
     +           ZETAY,ZETAZ
      INTEGER N,FLAG
      DIMENSION A(35),DA(35)


      DATA PI /3.141592/
C  NEGATIVE OF THE ZEROES OF THE AIRY FUNCTION
      DATA A /-2.33810, -4.08794, -5.52055, -6.78670, -7.94413,
     +         -9.02265,-10.04017,-11.00852,-11.93601,-12.82877,
     +        -13.69148,-14.52782,-15.34075,-16.13268,-16.90563,
     +        -17.66130,-18.40113,-19.12638,-19.83812,-20.53733,
     +        -21.22482,-21.90136,-22.56761,-23.22416,-23.87156,
     +        -24.51030,-25.14082,-25.76353,-26.37880,-26.98698,
     +        -27.58838,-28.18330,-28.77200,-29.35475,-29.93176/
C  THE DERIVATIVE OF THE AIRY FUNCTION EVALUATED AT THE ZEROES
      DATA DA/0.70121, -0.80311,  0.86520, -0.91085,  0.94733,
     +        -0.97792,  1.00437, -1.02773,  1.04872, -1.06779,
     +         1.08530, -1.10150,  1.11659, -1.13073,  1.14403,
     +        -1.15660,  1.16853, -1.17988,  1.19070, -1.20106,
     +         1.21098, -1.22052,  1.22970, -1.23854,  1.24708,
     +        -1.25534,  1.26334, -1.27109,  1.27861, -1.28592,
     +         1.29302, -1.29994,  1.30667, -1.31324,  1.31965/

C                      SET UNDERFLOW CONDITION TO NO PRINTOUT
```

```
      CALL ERRSET(208,0,-1,1,1,0)

C  FIND THE NORMALIZED EQUIVALENT OF RNG,Z1,Z2
              DO=(AE*AE*WAVE/1E5/PI)**.33333333
              HO=(AE*WAVE*WAVE/10/PI/PI)**.33333333/2
              X=RNG/DO
              Y=Z2/HO
              Z=Z1/HO

C        EXP (PI*I/3)
              EPI =   ( 0.5000001812D+00,  0.8660252991D+00)
C  COEF1 = 2*SQRT(PI)
              COEF1 = 0.3544908524D+01
C  COEF2 = 1/2*(SQRT(3)+I)
              COEF2 = ( 0.8660254478D+00,  0.5000000000D+00)
              COEF = COEF1 * DSQRT (X)
              SENSI= 0.0005
C  INITIALIZE
              FLAG = 0
              N = 1
              APTERM = 10000.
              SUM = (0.,0.)

C  COMPUTE 35 TERMS IN FOLK'S SERIES OR UNTIL THE CONTRIBUTION OF
C    OF TWO SUCCESSIVE TERMS IS LESS THAN 0.0001 FOR EACH ONE
2            IF ((N.EQ.36).OR.(FLAG.EQ.-2).OR.(FLAG.EQ.-1)) GOTO 1
                AY = A(N)+EPI*Y
                FNY = FAIRY(AY)/(EPI*DA(N))
                AZ = A(N)+EPI*Z
                FNZ = FAIRY(AZ)/(EPI*DA(N))
C     THE VALUES RETURNED BY FAIRY ARE AI(W)*EXP(ZETAW), WHERE
C        ZETAW = 2./3.*(W**(3./2.))
                ZETAY = 2./3.*(AY*CDSQRT(AY))
                ZETAZ = 2./3.*(AZ*CDSQRT(AZ))
                ETERM = -ZETAY-ZETAZ+(COEF2*A(N)*X)
                TERM = FNY*FNZ*CDEXP(ETERM)
                ATERM = CDABS (TERM)
                IF (ATERM.LE.10000) GOTO 3
C       FORCE THE PROGRAM TO TERMINATE IF ONE TERM CONTRIBUTES
C                       MORE THAN 1000
                 FLAG =-2
3               CONTINUE
                IF ((APTERM.GE.SENSI).OR.(ATERM.GE.SENSI)) GOTO 4
                  FLAG =-1
```

```
4               CONTINUE
                APTERM = ATERM
                SUM = SUM + TERM
                N = N+1
              GOTO 2
1             CONTINUE
              F = COEF *CDABS(SUM)
C  F=-1 WHEN 35 TERM ARE NOT ENOUGH I.E SPH35N "DIVERGED"
              IF (FLAG.NE.0) GOTO 5
                 F = -1.
5             CONTINUE
C  F=-2 WHEN ONE TERM CONTRIBUTES MORE THAN 10000
              IF (FLAG.NE.-2) GOTO 6
                  F = -2.
6             CONTINUE
              RETURN
              END


C******************************************************************
C                 FUNCTION TO CALCULATE AIRY FUNCTIONS          *
C                     IN THE COMPLEX PLANE                      *
C THE POWER SERIES AND GAUSSIAN QUADRATURE METHODS  ARE USED IN  *
C THE UPPER HALF PLANE. THE PROPERTY OF COMPLEX CONJUGATE OF     *
C AIRY FUNCTIONS IS USED TO CALCULATE THE AIRY FUNCTION IN THE   *
C LOWER PLANE.                                                   *
C                                                                *
C  * THIS FUNCTION RETURNS AI(Z)*EXP(2*(Z**3/2)/2)               *
C                                                                *
C******************************************************************
          COMPLEX FUNCTION FAIRY*16 (Z)


              REAL*8    RZ,IZ,TANZ
              COMPLEX*16 Z,AIR,AI,EPI,EPINT,EPIN,EPIT,AZETA


C  CALL ERRSET TO SUPPRESS MESSAGES WHEN ONE OF THE TERMS UNDERFLOWS
C   IN THE CONNECTION FORMULA
              CALL ERRSET (208,0,-1,1,1,0)


C  ASSIGN THE EXP NEEDED FOR THE CONNECTION FORMULA
```

```
C         EXP (PI*I/3)
              EPI =    ( 0.5000001812D+00,  0.8660252991D+00)
C         EXP (-2*PI*I/3)
              EPINT=  (-0.4999996375D+00, -0.8660256131D+00)
C         EXP (-PI*I/3)
              EPIN =  ( 0.5000001812D+00, -0.8660252991D+00)
C         EXP (-4*PI*I/3) = EXP (2*PI*I/3)
              EPIT =  (-0.4999996375D+00,  0.8660256131D+00)

              RZ = DREAL (Z)
              IZ = DIMAG (Z)


C  IF WE ARE IN THE CONNECTING FORMULA WEDGE  2PI/3<=ARG Z<=4PI/3
C    APPLY MAPPING FORMULA TO GET OUT OF THE WEDGE

              AZETA = CDEXP (4.*(Z*CDSQRT(Z))/3.)
              TANZ = IZ/RZ
C  AI(Z) RETURNS AIRY(Z)*EXP(2/3*Z**2/3)
              IF ((RZ.LT.0).AND.(TANZ.LE.1.7032).AND.
     +         (TANZ.GE.-1.7032)) GOTO 1
                AIR = AI (Z)
              GOTO 2
C  CONNECTION FORMULA FOR  AI(Z)
1               AIR = EPI*AI(Z*EPINT)*AZETA+EPIN*AI(Z*EPIT)
2             CONTINUE
              FAIRY = AIR
              RETURN
              END



          COMPLEX FUNCTION  AI*8 (AZ)

              COMPLEX*16 AZ,POWER,GQA
              REAL*8 MAZ,RAZ

C                          SET UNDERFLOW CONDITION TO NO PRINTOUT
      CALL ERRSET(208,0,-1,1,1,0)
              MAZ = CDABS (AZ)
              RAZ = DREAL (AZ)
C  IF  2PI/3 <=ARG ZA <= PI/2 OR 4PI/3 <= ARG ZA <= 3PI/2
               IF (RAZ .LE.0) GOTO 3
```

```
                IF (MAZ.LE.2) GOTO 7
                  AI = GQA (AZ)
                GOTO 8
7                 AI = POWER (AZ)
8               CONTINUE
               GOTO 4
3               IF (MAZ .LE. 4) GOTO 5
                   AI = GQA (AZ)
                GOTO 6
5                  AI = POWER (AZ)
6               CONTINUE
4              CONTINUE

C  IF IN THE RIGHT HALF PLANE ( REAL(ZA) >= 0)
              RETURN
              END


C*****************************************************************
C             POWER SERIES OF AIRY FUNCTION                       *
C POWER SERIES ARE COMPUTED ITERATIVELY UNTIL THE ADDITIONAL TERM*
C             CONTRIBUTES LESS THAN 1*E-10                        *
C                                                                 *
C THE FORMULA USED WAS TAKEN FROM J.C.MILLER "THE AIRY INTEGRAL",*
C BRITISH ASSOC. ADV. SCI., MATH TABLES VOLUME B,1946 B17         *
C*****************************************************************


          COMPLEX FUNCTION POWER*16 (Z)


              REAL*8 ALPHA,BETA,DEN1,DEN2,NUM1,NUM2,ATERM,FACTOR
     +           IN31, IN32
              COMPLEX*16 Z,AIRY,TERM1,TERM2,TERM,Y1,Y2,P1,P2,EZETA
              INTEGER*4 POWER1,POWER2,FP1,FP2,N

C                        SET UNDERFLOW CONDITION TO NO PRINTOUT
      CALL ERRSET(208,0,-1,1,1,0)
              ALPHA = 0.355028053887817
              BETA  = 0.258819403792807
              ATERM = 100000.
              EZETA = CDEXP (2.*(Z*CDSQRT(Z))/3.)
              N = 1
              AIRY  = ALPHA-(BETA*Z)
```

```
              POWER1= 3
              POWER2= 4
C  CONTINUE ADDING TERMS UNTIL THE CONTIBUTION OF THE LAST TERM IS
C  LESS THAN 1E-10
2             IF (ATERM.LE.1E-10) GOTO 1
              P1=Z**POWER1
              P2=Z**POWER2
              FP1 = POWER1
              FP2 = POWER2
              DEN1= FACTOR(FP1)
              DEN2= FACTOR(FP2)
              NUM1= IN31(N)
              NUM2= IN32(N)
              TERM1 = ALPHA*(NUM1/DEN1*P1)
              TERM2 = BETA *(NUM2/DEN2*P2)
              TERM = TERM1- TERM2
              AIRY = AIRY + TERM
              POWER1= POWER1+3
              POWER2= POWER1+1
              N = N+1
              ATERM = CDABS(TERM)
              GOTO 2
1             CONTINUE
              POWER = AIRY*EZETA
              RETURN
              END
C
C*******************************************************************
C---------FUNCTION THAT RETURNS THE FACTORIAL OF AN  INTEGER------*
C*******************************************************************
C
              REAL FUNCTION FACTOR*8(N)
C
C    SET UNDERFLOW CONDITION TO NO PRINTOUT
      CALL ERRSET(208,0,-1,1,1,0)
              IF ((N.EQ.0) .OR. (N.EQ.1)) GOTO 10
              FACTOR = 1.
30            IF (N.EQ.1) GOTO 20
              FACTOR = FACTOR*N
              N = N-1
              GOTO 30
20            RETURN
10            FACTOR=1.
              RETURN
```

```
           END
C
C
C*****************************************************************
C----------FUNCTION THAT EVALUATES 1*4*7....3*(N-1)+1------------*
C*****************************************************************
C
           REAL FUNCTION IN31*8(N)
C
C                     SET UNDERFLOW CONDITION TO NO PRINTOUT
      CALL ERRSET(208,0,-1,1,1,0)
           IN31 =1
           IF (N.EQ.1) GOTO 50
           DO 40 I=2,N
           IN31 = IN31*(3*(I-1)+1)
40         CONTINUE
50         RETURN
           END
C
C
C*****************************************************************
C----------FUNCTION THAT EVALUATES 2*5*8....3*(N-1)+2------------*
C*****************************************************************
C
           REAL FUNCTION IN32*8(N)
C
C                     SET UNDERFLOW CONDITION TO NO PRINTOUT
      CALL ERRSET(208,0,-1,1,1,0)
           IN32 =2
           IF (N.EQ.1) GOTO 60
           DO 70 I=2,N
           IN32 = IN32*(3*(I-1)+2)
70         CONTINUE
60         RETURN
           END


C*****************************************************************
C          PROCEDURE TO CALCULATE THE AIRY FUNCTION OF           *
C   A COMPLEX ARGUMENT BY THE GAUSSIAN QUADRATURES  METHOD       *
C    AI(Z) = 1/2*(PI**(-1/2))*(Z**(-1/4))*EXP(ZETA)*SUM OVER N   *
C               W(I)/(1+X(I)/ZETA))                              *
C                                                                *
```

```
C    IMPLEMENTED AS IN "AN ALGORITHM FOR THE EVALUATION OF          *
C      OF COMPLEX AIRY FUNCTIONS", JOURNAL OF COMPUTATIONAL         *
C               PHYSICS 31, 60-75 (1979)                            *
C                                                                   *
C     THIS FUNCTION RETURNS AI(Z)*EXP(ZETA)                         *
C                                                                   *
C     WEIGHTS AND ZEROES WERE CALCULATED BY GQAIRY.PLI              *
C*******************************************************************


              COMPLEX FUNCTION GQA*16 (Z)

              REAL *8 ZEROES (10),WEIGHT(10)
              COMPLEX *16 AIR,Z,SUM,ZETA
              INTEGER N
              DATA DSQRPI /0.564189584/

C SET UNDERFLOW CONDITION TO NO PRINTOUT
      CALL ERRSET(208,0,-1,1,1,0)


C INITIALIZE (ASSIGN VALUES TO WEIGHTS AND X-INTERCEPT)

              ZEROES(1)= 1.408308107197377E+01
              ZEROES(2)= 1.021488548060315E+01
              ZEROES(3)= 7.441601846833691E+00
              ZEROES(4)= 5.307094307915284E+00
              ZEROES(5)= 3.634013504378772E+00
              ZEROES(6)= 2.331065231384954E+00
              ZEROES(7)= 1.344797083139945E+00
              ZEROES(8)= 6.418885840366331E-01
              ZEROES(9)= 2.010034600905718E-01
              ZEROES(10)= 8.059435921534400E-03
              WEIGHT(1)= 2.677084371247434E-14
              WEIGHT(2)= 6.636768688175870E-11
              WEIGHT(3)= 1.758405638619854E-08
              WEIGHT(4)= 1.371239148976848E-06
              WEIGHT(5)= 4.435096659959217E-05
              WEIGHT(6)= 7.155501075431907E-04
              WEIGHT(7)= 6.488956601264211E-03
              WEIGHT(8)= 3.644041585109798E-02
              WEIGHT(9)= 1.439979241604145E-01
              WEIGHT(10)= 8.123114134235980E-01
```

```
      SUM = 0.
      ZETA = 2.*(Z*CDSQRT(Z))/3.

      DO 100 I=1, 10

        SUM = SUM + (WEIGHT(I)/(1.0+(ZEROES(I)/ZETA)))
100   CONTINUE

      GQA = 0.5*DSQRPI/(CDSQRT(CDSQRT(Z)))*SUM

      RETURN
      END
```

UNCLASSIFIED

SECURITY CLASSIFICATION OF THIS PAGE

REPORT DOCUMENTATION PAGE

1a. REPORT SECURITY CLASSIFICATION Unclassified	1b. RESTRICTIVE MARKINGS
2a. SECURITY CLASSIFICATION AUTHORITY	3. DISTRIBUTION/AVAILABILITY OF REPORT Approved for public release; distribution unlimited.
2b. DECLASSIFICATION/DOWNGRADING SCHEDULE	
4. PERFORMING ORGANIZATION REPORT NUMBER(S) CMT-111	5. MONITORING ORGANIZATION REPORT NUMBER(S) ESD-TR-88-122
6a. NAME OF PERFORMING ORGANIZATION Lincoln Laboratory, MIT 6b. OFFICE SYMBOL *(If applicable)*	7a. NAME OF MONITORING ORGANIZATION Electronic Systems Division
6c. ADDRESS *(City, State, and Zip Code)* P.O. Box 73 Lexington, MA 02173-0073	7b. ADDRESS *(City, State, and Zip Code)* Hanscom AFB, MA 01731
8a. NAME OF FUNDING/SPONSORING ORGANIZATION Air Force Systems Command, USAF 8b. OFFICE SYMBOL *(If applicable)*	9. PROCUREMENT INSTRUMENT IDENTIFICATION NUMBER F19628-85-C-0002
8c. ADDRESS *(City, State, and Zip Code)* Andrews Air Force Base Washington, DC 20334	10. SOURCE OF FUNDING NUMBERS PROGRAM ELEMENT NO.: 63003F, 63001F PROJECT NO.: 331 TASK NO.: WORK UNIT ACCESSION NO.:

11. TITLE *(Include Security Classification)*
An Improved Spherical Earth Diffraction Algorithm for SEKE

12. PERSONAL AUTHOR(S)
Michael P. Shatz and George H. Polychronopoulos

13a. TYPE OF REPORT	13b. TIME COVERED	14. DATE OF REPORT *(Year, Month, Day)*	15. PAGE COUNT
Project Report	FROM ______ TO ______	1988, April, 15	52

16. SUPPLEMENTARY NOTATION
None

17. COSATI CODES			18. SUBJECT TERMS *(Continue on reverse if necessary and identify by block number)*
FIELD	GROUP	SUB-GROUP	spherical earth diffraction; SEKE; radar propagation; Airy function; Fock's series; polynomial fit

19. ABSTRACT *(Continue on reverse if necessary and identify by block number)*

The spherical earth diffraction subroutine SPH35 in the radar propagation code SEKE has been known to cause errors in propagation loss computations for a range of combinations of antenna and target heights. In this report an efficient method to evaluate the Airy function in the complex plane is presented. This method uses the power series expansion near the origin and an integral representation elsewhere. It is more accurate and as fast as the method employed in the spherical earth diffraction subroutine SPH35 that evaluates every Airy function of Fock's series by a fourth-order polynomial fit to its logarithm. The algorithm presented was incorporated in a new spherical earth diffraction subroutine (SPH35N). It was found that, if SEKE uses this subroutine, no problems arise for normalized heights of up to 5000 (i.e. about 350 km at VHF or 17 km at K_u band).

The subroutine SPH35N, described in this report, has been used in the versions of SEKE running at Lincoln Laboratory, and is in the version of SEKE currently being supplied to other users.

20. DISTRIBUTION/AVAILABILITY OF ABSTRACT ☐ UNCLASSIFIED/UNLIMITED ☒ SAME AS RPT. ☐ DTIC USERS	21. ABSTRACT SECURITY CLASSIFICATION Unclassified
22a. NAME OF RESPONSIBLE INDIVIDUAL Lt. Col. Hugh L. Southall, USAF	22b. TELEPHONE *(Include Area Code)*: (617) 981-2330 22c. OFFICE SYMBOL: ESD/TML